

```

*****
*                                     *
*   T   E   I   L       6           *
*                                     *
*****
*                                     *
*   G   R   A   F   I   K           *
*                                     *
*   U   N   D                   *
*                                     *
*   F   A   R   B   E   N           *
*                                     *
*****

```

```

*   GRAFIKZEICHEN
*   *****
*   ZEICHENBEWEGUNG (TRICK)
*   *****
*   STEUERUNG DER FARBEN
*   *****
*   HOCHAUFLÖSENDE GRAFIK
*   *****
*   PUNKTE, LINIEN UND ÜBERSCHRIFTEN
*   *****
*   QUADRATE, KREISE, VIELECKE
*   UND MALEREI
*   *****

```

* 6.1 GRAFIKZEICHEN *

Im Teil 2 dieses Handbuchs wurde bereits besprochen, daß hinter jeder Buchstabentaste zusätzliche Grafikzeichen stehen. Gleiches bei den Tasten <+>, <->, <*>, <@>, <=> und dem Pfund-Zeichen, wodurch sich insgesamt 62 verschiedene Grafikzeichen ergeben. Um diese Grafikzeichen auszudrucken, müssen entweder die Taste <Shift> oder die COMMODORE-Taste <C=> und gleichzeitig eine der oben beschriebenen Tasten - mit dem gewünschten Symbol - niedergedrückt werden.

Befindet sich der COMMODORE 116 nicht im Schreibmaschinen-Modus, so wird mit Taste <Shift> und der Buchstabentaste das r e c h t e Grafikzeichen gedruckt. Bereits dieser Grafiksatz enthält alle Spielkartensymbole, volle und leere Ballsymbole und eine ganze Reihe sich ergänzender Kurven und Linien. Dieser Satz reicht schon für vielfältigste Anwendungen.

Wird die COMMODORE-Taste <C=> in Verbindung mit einer Buchstabentaste gedrückt, so erscheinen die l i n k e n Grafikzeichen. Diese umfassen insgesamt Balken, Quadrate, Linien und Blöcke, um grafische Darstellungen und Diagramme zu zeichnen.

WIE MAN SPIELKARTEN ENTWIRFT *****

Nachstehend wird erklärt, wie vielfältig und sinnvoll sich diese Grafiksymbole einsetzen lassen. Folgen Sie diesen Instruktionen und entwerfen Sie eine richtige Spielkarte, eine 'HERZ-SECHS'.

Als erstes wird die CURSOR-Farbe in ROT geändert. Taste <Control> gedrückt halten und die Zifferntaste <3> antippen - und der CURSOR blinkt in Rot.

Nächster Schritt ist die Oberkante der Spielkarte. Da in der Folge nur Grafikzeichen verwendet werden, wird die Taste <Shift> bei allen Zeichen niedergehalten, der CURSOR etwa zehn bis zwölf Positionen vom linken Rand entfernt und aufeinanderfolgend:

1 x Taste <U>, 5 x Taste <C> und 1 x Taste <I>
gedrückt. Nachstehend das Ergebnis:

(-----)

Nun die linke Spielkartenseite. Die Taste <Shift> wird wiederum bei allen Zeichen gedrückt, und der CURSOR muß in das Feld unter dem linken Bogen (der durch die Taste <U> entstand). Mit der Taste werden ab dieser Position - nach unten - sieben senkrechte Striche gezogen. (Dabei üben Sie auch den Umgang mit dem CURSOR-Kreuz):

┌-----┐
│
│
│
│
│
│
│


Die Unterkante und die rechte Spielkartenseite sind genauso einfach zu zeichnen:

1 x Taste <J>, 5 x Taste <C>, 1 x Taste <K> und 5 x
gedrückt ... und nachstehend das Ergebnis:



Jetzt folgen die Spielkarten-Einträge. Dazu den CURSOR zuerst in die linke, obere Spielkartenecke bewegen und eine '6' eintragen (Taste <6>); und in der rechten, unteren Ecke (eine '6' spiegelbildlich) eine '9'.

Letzter Schritt ist dann die Plazierung der Herz-Symbole, und zwar sechsmal. Diese Einträge müssen symmetrisch auf der Spielkarte verteilt sein - etwa so:



Keine allzu gute Karte etwa im Kartenspiel 'Black Jack', aber Sie können ja weiterüben und durch diese Praxis Ihre Fähigkeit erhöhen.

ANMERKUNG: Bei gedrückter Taste <Shift> kann die Taste <Return> ohne Gefahr eines 'SYNTAX ERROR' gedrückt werden, egal was in der Zeile des CURSORS auch geschrieben steht. Der COMMODORE 116 interpretiert in diesem Fall die Taste <Return> nicht der BASIC-Sprache entsprechend als Startbefehl, sondern lediglich als Aufforderung, den CURSOR an den Anfang der nächsten Zeile zu bringen.

Mit Hilfe der Grafik-Tasten läßt sich der Ausdruck wesentlich schöner und interessanter gestalten. Das nächste Beispiel zeigt, wie Worte oder Spalten unterstrichen werden können.

Den CURSOR als erstes in die nächste Zeile (unter das erste zu unterstreichende Zeichen) bewegen. Dann die COMMODORE-Taste <C=> gedrückt halten und die Taste <T> ebenfalls drücken - solange bis die gewünschte Unterstrichlänge erreicht ist. Beide Tasten wieder loslassen.

Zweck dieses Abschnitts war es, die vielfältigen Grafikmöglichkeiten des COMMODORE 116 aufzuzeigen und diese bei der Gestaltung unterschiedlichster Formen und Figuren einzusetzen. Zusätzlich zu den 62 Grafikzeichen steht diese Anzahl nochmals in REVERS-Darstellung zur Verfügung (siehe die Beschreibung der 'Rennstreifen' - Kapitel 4.3). Wie immer macht Übung auch hier den Meister.

```
*****
* 6.2 ZEICHENBEWEGUNG (TRICK) *
*****
```

Jeder Film besteht aus einer Vielzahl von Einzelbildern, und jedes Bild unterscheidet sich ein wenig vom vorhergehenden. Der Projektor wirft jedes dieser Bilder nur für eine Sekundenbruchteil auf die Leinwand, und schon kommt das nächste. Durch die Trägheit unseres Auges entsteht für den Betrachter der Eindruck eines bewegten Bildes.

Computer-Bewegung funktioniert genauso. Zuerst bringt der Computer ein Bild, dann wird das Bild ein wenig geändert. Der COMMODORE 116 ist schnell genug (der Bildwechsel muß mindestens 25 mal in der Sekunde erfolgen, um fließende Bewegungen zu erzielen), um den Effekt sich langsam bewegendere Objekte auf den Bildschirm - sowohl in Spielen wie in Programmen der Praxis - 'zaubern' zu können.

Programme zur Erstellung solcher Bilder verwenden die PRINT-Anweisung in Verbindung mit den Grafikzeichen. Einfachstes Beispiel eines Bewegungsablaufes sind zwei sich abwechselnde Zeichen.

Ein erstes Beispielprogramm läßt abwechselnd einen Kreis (Taste <Shift> und gleichzeitig Taste <Q> drücken) und ein Herz (Taste <Shift> und gleichzeitig Taste <S> drücken) erscheinen. Mit einiger Phantasie läßt sich dies bereits als 'Herzschlag' interpretieren.

```
10 PRINT "◻●":FOR I = 1 TO 200:NEXT
    ! !
    ! Tasten <Shift> & <Q>
    Tasten <Shift> & <Clr/Home>
```

Das Verbindungs-
zeichen '&' be-
deutet 'gleich-
zeitig drücken.'

```

20 PRINT "♠♥":FOR I = 1 TO 200:NEXT
      !!
      ! Tasten <Shift> & <S>
      Tasten <Shift> & <Clr/Home>
30 GOTO 10

```

Das Verbindungs-
zeichen '&' be-
deutet 'gleich-
zeitig drücken'.

WICHTIG: Für den weiteren Beschreibungsteil wird darauf hingewiesen, daß das Verbindungszeichen '&' den gleichzeitigen Druck der mit diesem Zeichen verbundenen Tasten bedeutet. Die jeweils zuerst genannte Taste ist immer etwas vor der nächsten zu betätigen.

Vor jedem neuen Programm - ob es nun abgespeichert wurde oder nicht - ist es ratsam, 'NEW' einzugeben und die Taste <Return> zu drücken (Speicher wieder in vollem Umfang frei). Jede Programmzeile muß ebenfalls mit der Taste <Return> abgeschlossen werden (damit übernimmt der Computer diese Zeile in den Speicher).

```

10 PRINT "♠♠"
      !!
      ! Tasten <Shift> & <A>
      Taste <Clr/Home>

```

Das Verbindungs-
zeichen '&' be-
deutet 'gleich-
zeitig drücken'.

```

20 FOR L = 1 TO 100
30 NEXT L
40 PRINT "♠♣"
      !!
      ! Tasten <Shift> & <X>
      Taste <Clr/Home>

```

Das Verbindungs-
zeichen '&' be-
deutet 'gleich-
zeitig drücken'.

```

50 FOR M = 1 TO 100
60 NEXT M
70 GOTO 10

```

Die Grenzen dieser Bewegungsabläufe (nach dem RUN-Befehl) - mit zwei Zeichen - sind klar erkennbar. Gestoppt wird das Programm mit der Taste <Run/Stop>.

Interessantere Effekte lassen sich mit aus mehreren Grafiken zusammengesetzten Bildern erzielen, bei denen im zweiten Bild einige wenige Zeichen geändert werden, die anderen hingegen ihre Position behalten. Dadurch entsteht der Eindruck, als würde sich ein größeres Objekt bewegen. Das nachstehende Beispiel soll dies demonstrieren:

```

5 SCNCLR
10 PRINT "■ \ O /"
      ! ! ! !
      ! ! ! Tasten <Shift> & <N>
      ! ! ! Tasten <Shift> & <W>
      ! ! ! Tasten <Shift> & <M>
      Taste <Clr/Home>

20 PRINT "  ■  "
      ! ! !
      ! ! ! 1x LEER-Taste <SPACE>
      ! COMMODORE-Taste <C=> & <+>
      1x LEER-Taste <SPACE>

30 PRINT "/ \ "
      ! ! !
      ! ! ! Tasten <Shift> & <M>
      ! ! ! 1x LEER-Taste <SPACE>
      Tasten <Shift> & <N>

40 FOR L = 1 TO 100 : NEXT L
50 PRINT "■ O "
      ! ! ! !
      ! ! ! 1x LEER-Taste <SPACE>
      ! ! ! Tasten <Shift> & <W>
      ! ! ! 1x LEER-Taste <SPACE>
      Taste <Clr/Home>

60 PRINT " - ■ - "
      ! ! !
      ! ! ! COMMODORE-Taste <C=> & <T>
      ! ! ! COMMODORE-Taste <C=> & <+>
      COMMODORE-Taste <C=> & <T>

70 PRINT " | | "
      ! ! !
      ! ! ! COMMODORE-Taste <C=> & <G>
      ! ! ! COMMODORE-Taste <C=> & <G>
      1x LEER-Taste <SPACE>

80 FOR L = 1 TO 100 : NEXT L
90 GOTO 10

```

'RUN' eintippen und Taste <Return> drücken.

In den bisher gebrachten Bewegungsbeispielen blieb die bewegte Grafik an einem festen Ort des Bildschirms. In der nächsten Übung wird sich die Figur über den Bildschirm bewegen. Mit Hilfe der TAB-Funktion können Objekte von der linken Bildschirmkante weg bewegt werden (Details zur TAB-Funktion finden Sie im BASIC-Lexikon). Im folgenden Beispiel wird sich eine Schlange über den Bildschirm bewegen.

Und vergessen Sie nicht, daß die mit '&' verbundenen Tasten gleichzeitig gedrückt werden müssen.

```

5 FOR A = 0 TO 30
10 SCNCLR
20 PRINT TAB(A) "<Shift>&<U> <Shift>&<I> <Shift>&<U> <Shift>&<I>"
30 PRINT TAB(A) "<Shift>&<K> <Shift>&<J> <Shift>&<K> <Shift>&<J>"
40 FOR L = 1 TO 100 : NEXT L
50 SCNCLR
60 PRINT TAB(A+1)"<Shift>&<I> <Shift>&<U> <Shift>&<I> <Shift>&<U>"
70 PRINT TAB(A+1)"<Shift>&<J> <Shift>&<K> <Shift>&<J> <Shift>&<K>"
80 FOR L = 1 TO 100 : NEXT L
90 NEXT A

```

Mit dem Grafikzeichen "Ball" (Tasten <Shift>&<Q>) lassen sich bereits Videospiele auf dem Schirm spielen. Um den Ball zu bewegen, genügt es, ihn zu löschen und an der neuen Position wieder darzustellen - wie nachstehendes Programm zeigt:

```

10 PRINT " <Shift> & <Clr/Home> "
20 PRINT "  ● ■■ ";
    ! ! !
    ! ! Taste <CRSR/Links>
    ! Tasten <Shift> & <Q>
    1x LEER-Taste <SPACE>

30 FOR L = 1 TO 50 : NEXT L
40 GOTO 20

```

* 6.3 STEUERUNG DER FARBEN *

An jede Stelle des Bildschirms kann eine andere Farbe gesetzt werden. So kann z.B. die Einrahmung des Bildschirms in einer, der Hintergrund in einer zweiten und jeder Buchstabe in einer weiteren eigenen Farbe sein. Wie die einzelnen Buchstabenfarben zu setzen sind, wurde bereits erklärt (Kapitel 2.3). Farbänderungen von Rand und Hintergrund erfolgen über die BASIC-Anweisung COLOR.

Durch die Anweisung COLOR 4,3 und Drücken der Taste <Return> wird z.B. der Bildschirmrand rot. Mit '4' wird in dieser Anweisung der Randbereich und mit der '3' die Farbe Rot angesprochen (Taste <3>).

Bei der Anweisung COLOR 0,7 und der Taste <Return> wechselt die Hintergrundfarbe in die Farbe Blau. Dabei stellt die Null ('0') die Adresse für den Hintergrund dar und die Sieben ('7') für die Farbe Blau (repräsentiert durch die Taste <7>).

Somit können wir festhalten:

die erste Zahl hinter der Anweisung COLOR steht für den Bildschirmbereich, der geändert werden soll. In nachstehender Tabelle wird ein Gesamtüberblick gegeben. Die Bereiche 2 und 3 werden im Kapitel 6.7 Mehrfarbengrafik besprochen.

NUMMERN DER BILDSCHIRMBEREICHE

Bereichs-Nr.	Bereichs-Bezeichnung
0	Bildschirm-Hintergrund
1	Buchstaben, Zeichen (Vordergrund)
2	Mehrfarben 1
3	Mehrfarben 2
4	Bildschirm-Rand

Die zweite Zahl in der COLOR-Anweisung selektiert die Farbe des gewählten Bildschirmbereichs. Diese Zahl ist mit den Farbtasten der Tastatur identisch.

FARBNUMMERN

Farb-Nr.	Farbe	Farb-Nr.	Farbe
1	Schwarz	9	Orange
2	Weiß	10	Braun
3	Rot	11	Glb/Grün
4	Zyan	12	Rosa
5	Purpur	13	Bla/Grün
6	Grün	14	Hl/Blau
7	Blau	15	Dk/Blau
8	Gelb	16	Hl/Grün

Jede Farbe ist auch noch in ihrer Helligkeit, der 'LUMINANZ' veränderbar. Im Anhang an die Farb-Nummer kann diese LUMINANZ von Null ('0'= dunkel) bis sieben ('7'= hell) variieren. Geben Sie COLOR 4,3,0 ein, und der Bildschirmrand wird sich in dunklem Rot präsentieren. Bei COLOR 4,3,7 erscheint die Einrahmung in hellem Rot.

Zusammenfassend lautet die COLOR-Anweisung:

COLOR (Bereich), (Farbe), (Helligkeit)

Mit dem nachstehenden kurzen Programm werden Ihnen alle 16 Commodore-Farben gezeigt. Zuerst NEW eingeben und Taste <Return> drücken:

```

5 SCNCLR
10 COLOR 0, 7, 7
20 FOR M=0 TO 7
30 FOR N=1 TO 2
40 FOR L=1 TO 16
50 PRINT "R ";
      !
      Taste <Control> & <Rvs/On>

60 READ A
70 COLOR 1, A, M
80 PRINT " ";
      ! !
      ! !
      ! 1x LEER-Taste <SPACE>
      1x LEER-Taste <SPACE>

90 NEXT L
100 PRINT
110 RESTORE
120 NEXT N,M
130 COLOR 1, 2, 4
200 DATA 7,14,4,13,6,16,11,8,10,9,3,12,5,15,2,1

```

Wird dieses Programm mit RUN gestartet, so verändert sich die Hintergrundfarbe des Bildschirms in Hellblau, und das Spektrum der 16 Commodorefarben wird im gesamtmöglichen Helligkeitsbereich (0-7) gezeigt. Dabei werden Sie feststellen, daß die Farbe Schwarz im gesamten Helligkeitsspielraum schwarz bleibt.

WICHTIG: Wie für die anderen in diesem Kapitel wiederholten BASIC-Grafik-Kommandos auch, gilt für das BASIC-Wort COLOR, daß es sowohl als Anweisung wie auch als Kommando vorkommen kann. Dabei bleibt es solange unwichtig, dies besonders hervorzuheben (gleiches gilt für die anderen Grafik-Befehle), bis klargestellt ist, welche Formulierung im DIREKT-Modus oder im PROGRAMM-Modus öfter vorkommt.

GRAFIK-MODUS *****

Die bisher gezeigten Grafiksymbole sind Einzelsymbole der Tastatur und stehen in keinem Verhältnis zu den weiteren Möglichkeiten des COMMODORE 116. Im Befehlsvorrat des COMMODORE-116 - BASICs sind Befehle enthalten, die es ermöglichen, Grafiken auch per Programm zu erzeugen. Um diese grafikbezogenen Befehle einsetzen zu können, muß in den GRAFIK-Modus umgeschaltet werden.

Der GRAFIK-Modus kann gleichzeitig als ZEICHNUNGS-Modus verstanden werden, da auch alle für das Zeichnen nötigen Befehle aktiviert sind. Bevor der entsprechende GRAFIK-Modus - per GRAPHIC-Befehl - nicht ausgewählt ist, sind die oben erwähnten Befehle nicht einsetzbar.

Wir unterscheiden drei GRAFIK-Modi: Normaler TEXT, Hochauflösende GRAFIK und Mehrfarbige GRAFIK. Per GRAPHIC-Befehl ist auch die Kombination der Modi möglich, wobei z.B. ein Teil des Bildschirms mit Text beschrieben wird, während auf dem anderen Teil gezeichnet werden kann. Der BASIC-Befehl für diesen Modus lautet GRAPHIC.

Die Syntax des 'Graphic'-Kommandos:

GRAPHIC (Modus), (Bildschirm löschen Ja, Nein)

Modus	Grafik-Modus
0	Text
1	Hochauflösende Grafik
2	Hochauflösende Grafik und Text
3	Mehrfarben-Grafik
4	Mehrfarben-Grafik und Text

Bildschirm löschen	Modus
0	Bildschirm wird n i c h t gelöscht
1	Bildschirm wird gelöscht

Um vom normalen Grafik-Modus (dem Text-Modus) in den Modus der Hochauflösenden Grafik umzuschalten, ist nur die Anweisung GRAPHIC 2,1 und die Taste <Return> erforderlich. Daraufhin wird der Bildschirm gelöscht, und der CURSOR befindet sich am unteren Bildschirmende. Der COMMODORE 116 hat nun einen in zwei Bereiche geteilten Bildschirm:

Der obere Bildschirmbereich ist für Grafik vorbereitet, und die letzten fünf Zeilen sind für Texte reserviert.

(Werden diese fünf Textzeilen nicht gewünscht, dann lautet der Befehl GRAPHIC 1,1 - dieser Modus hat jedoch den Nachteil, daß auch keiner der eingetippten Befehle zu lesen ist!)

Mit der GRAPHIC-Anweisung ist auch das fortlaufende Hin- und Herschalten zwischen GRAFIK- und TEXT-Modus möglich. Zum TEXT-Modus wird mit der Anweisung GRAPHIC 0 und wieder in den Hochauflösenden GRAFIK-Modus durch GRAPHIC 2 geschaltet, ohne dabei den Bildschirm zu löschen.

Es gibt noch eine Möglichkeit, im Hochauflösenden Grafik-Modus den Bildschirm zu 'clearen'. Mit dem Befehl SCNCCLR (SCreenNCLeaR) wird der Bildschirm gelöscht, ohne daß der Grafik-Modus verlassen wird. Nach Umschaltung in den Hochauflösenden Grafik-Modus werden 10 KByte des BASIC-Programmspeicher-Bereichs dafür abgetrennt. Wird ausschließlich mit Grafik gearbeitet, so besteht durch den Befehl GRAPHIC CLR die Möglichkeit, diesen abgetrennten Speicherbereich zurückzugewinnen.

```
*****
* 6.4 HOCHAUFLÖSENDE GRAFIK *
*****
```

Die Bildschirmaufteilung des COMMODORE 116 setzt sich aus 25 Zeilen mit je 40 Zeichenpositionen (= insgesamt 1000 Positionen) zusammen. Jedes Zeichen wird aus einer Punktmatrix von 8 Reihen mit je 8 Punkten (= 64 Punkte/Zeichen) gebildet. Pro Bildschirmzeile sind dies 40 Positionen x 8 Punkte pro Matrix-Reihe (= 320 Punkte). Vertikal ergeben sich somit 8 Punkte pro Zeile (= 200 Punkte). In Hochauflösender Grafik sind dies insgesamt 320 x 200 Punkte (= 64 000 Punkte), die beim COMMODORE 116 einzeln angesprochen werden können.

In x-Richtung (Spalten) dürfen Koordinaten-Zahlenwerte von 0 bis 319 vorkommen, in y-Richtung (Zeilen) Werte von 0 bis 199. Dabei entspricht y = 0 der obersten Punktreihe!

Im normalen Grafik-Modus ist der Zugriff auf das gesamte Zeichen begrenzt. Dabei mag die Kreation eines Rennstreifens oder einer Spielkarte ein hübsches Beispiel sein, doch die Möglichkeiten bleiben auf die den einzelnen Tasten zugeordneten Symbole beschränkt.

Zwar ergeben sich dabei genügend Möglichkeiten, um die verschiedensten Formen und Figuren zu entwerfen, doch eben nur ein Bruchteil der Möglichkeiten bei Einzelpunktansteuerung. Und diese erweiterte Möglichkeit haben Sie beim COMMODORE 116 im Hochauflösenden Grafik-Modus. Die erzielte Auflösung zeigt sich in der Genauigkeit und den Steuerungsmöglichkeiten beim Zeichnen; in Hochauflösender Grafik lassen sich Punkte, Linien, Kreise und andere Formen durch einfache Befehle zeichnen und wieder löschen.

Es gibt im Hochauflösenden Grafik-Modus (kurz: Hi-Res) eine Einschränkung: Pro 8×8 - Zeichen-Position sind nur zwei Farben darstellbar. Dies bedeutet, daß jede Zeichenposition des Bildschirms auf zwei Farben begrenzt ist (Vorder- und Hintergrundfarbe pro Zeichenfeld). Von Zeichenposition zu Zeichenposition kann die Farbe im gesamten Spektrumsbereich wechseln - innerhalb der Zeichenposition nur zwischen zwei Farben.

Im später in diesem Teil beschriebenen Mehrfarben-Grafik-Modus ist es möglich, bis zu vier Farben pro Zeichenposition zu definieren. Dies geht jedoch auf Kosten der Auflösung im Hi-Res-Grafik-Modus.

Das folgende Programmbeispiel zeigt einige der Hi-Res-Möglichkeiten des COMMODORE 116 in Verbindung mit dem Befehl GRAPHIC auf. Löschen Sie dafür mit dem NEW-Befehl den Speicher und tippen dann ein:


```

10 COLOR 0,1
20 GRAPHIC 1,1
30 FOR L=2 TO 16
40 COLOR 1,L,4
50 DRAW 1,0,L*12 TO 319,L*12
60 DRAW 1,L*18,0 TO L*18,199
70 NEXT L
80 FOR L=1 TO 5000 : NEXT
90 COLOR 1,2,7
100 GRAPHIC 0

```

Sie werden an den Übergängen eine gewisse Farbverfälschung feststellen. Diese ergibt sich aus den Grenzen der Hi-Res-Grafik, wenn zu viele Farben auf zu engem Raum zusammentreffen.

```

*****
* 6,5 PUNKTE, LINIEN UND ÜBERSCHRIFTEN *
*****

```

Geben Sie die Befehlsfolge GRAPHIC 2,1:DRAW 1,0,0 ein und drücken Sie die Taste <Return>. Und beobachten Sie genau die linke, obere Bildschirmecke. Der COMMODORE 116 hat mit dieser Befehlsfolge einen einzigen, schwarzen Punkt dorthin gesetzt. Mit dem Befehl DRAW können Einzelpunkte, Linien oder ganze Formen an jede Stelle des Bildschirms gezeichnet werden. Hier einige der DRAW-Befehle:

Befehl	Ergebnis
DRAW (Farbquelle), (Spalte),(Zeile)	PUNKT
DRAW (Farbquelle), (Spalte),(Zeile)TO(Spalte),(Zeile)	LINIE
DRAW (Farbquelle)TO(Spalte),(Zeile)	ZEICHNET LINIE AB DEM LETZTEN PUNKT

Farbquelle Null ('0') steht für den Farb h i n t e r grund des Zeichens, Eins ('1') für die Farbe des Vordergrund. Wird ein Punkt in der Hintergrundfarbe gesetzt, ist dies gleichbedeutend mit Löschung der Vordergrundfarbe an dieser Punktstelle.

In der Anweisung DRAW kann die erste Zahl entweder 1 (zeichne einen Punkt) oder 0 (lösche einen Punkt) sein. Die nächsten beiden Zahlen geben die Spalten- und Zeilenposition des Punktes an. Soll ein Punkt in Spalte 17, Zeile 20 gesetzt werden, dann lautet der Befehl DRAW 1,17,20. Gelöscht wird dieser Punkt wieder mit DRAW 0,17,20.

Mit der DRAW-Anweisung lassen sich auch Linien zwischen Punkten ziehen. Durch Hinzunahme des Befehls TO und der zweiten Punktposition: DRAW 1,1,1 TO 100,100. Damit wird eine Linie von dem Punkt in Position 1,1 nach dem Punkt in Position 100,100 gezogen. Diese Linie wird wieder gelöscht, wenn der Befehl lautet: DRAW 0,1,1 TO 100,100.

Wird diese Zeichentechnik auch im mathematischen (oder geometrischen) Bereich eingesetzt, so ist ein gewisser Umdenkprozeß notwendig. Im Gegensatz zum herkömmlichen Koordinatensystem (bei dem die Position 0,0 entweder in der Bildmitte oder links unten zu finden ist) beginnt der COMMODORE 116 in der linken oberen Ecke mit seiner Positionszählung. Bei einiger Übung wird dies für Sie jedoch kein Problem darstellen.

Ist erst einmal der erste Punkt oder die erste Linie am Bildschirm gezeichnet, so kann die Zeichnung von da fortgesetzt werden: DRAW 1 TO 150,50 bewirkt eine Linie - vom letzten Punkt aus - zur Position Spalte 150, Zeile 50. Werden in einem Programm eine Vielzahl DRAW TO-Befehle verwendet, so empfiehlt sich der Einsatz eines weiteren Befehls: LOCATE.

Mit dem Befehl LOCATE wird der Anfangspunkt an eine Bildschirmposition fixiert und damit die unverzügliche Rückkehr zu diesem Punkt erreicht: z.B. LOCATE 100,100.

NEW und Taste <Return>

```
10 COLOR 0,1
20 COLOR 1,2
30 GRAPHIC 1,1
40 LOCATE 0,100
50 FOR X=1 TO 319
60 Y = INT (100+99*SIN(X/20))
70 DRAW 1 TO X,Y
80 NEXT X
90 FOR L=1 TO 5000
100 NEXT L
110 GRAPHIC 0
```

Geben Sie nach der Erprobung des obigen Programms **k e i n** NEW ein. Um eine weitere Version laufen zu lassen, ändern Sie lediglich Zeile 70 in:

```
70 DRAW 1,X,Y
```

Mit dieser Änderung wird das Programm nicht mit Linien sondern mit Punkten zeichnen.

DIE CHAR-ANWEISUNG *****

Grafische Darstellungen werden übersichtlicher, wenn sie zusätzliche Beschriftung erhalten. Mit der Anweisung CHAR kann Hi-Res-Grafik mit Text gemischt werden. Die Anweisung CHAR 1,0,5,"ZEITACHSE" schreibt den Text 'ZEITACHSE' linksbündig in die 6.(!) Zeile des Bildschirms. Die erste Zahl in der CHAR-Anweisung entscheidet wiederum über Darstellen (= 1) oder Löschen (=0). Die folgenden zwei Zahlen bestimmen Spalte und Zeile, ab der der Text erscheint.

Belassen Sie eines der beiden letzten Programme im Computer; geben Sie weder NEW noch <Return> ein, sondern fügen Sie nachstehende Programmzeilen hinzu:

```
81 CHAR 1,0,0,"GRAFISCHE DARSTELLUNG DER":CHAR 1,0,1,"FORMEL"
82 CHAR 1,0,2,"Y = SIN(X)"
83 DRAW 1,0,100 TO 319,100,189,0 TO 189,199
84 CHAR 1,0,12,"X-ACHSE": CHAR 1,22,0,"Y"
85 CHAR 1,22,2,"A": CHAR 1,22,3,"C"
86 CHAR 1,22,4,"H": CHAR 1,22,5,"S"
87 CHAR 1,22,6,"E"
```

```
*****
* 6.6 QUADRATE, KREISE, VIELECKE UND MALEREI *
*****
```

RECHTECKE ZEICHNEN

```
*****
```

Unter Verwendung des DRAW-Befehls lassen sich Bilder durch Aneinanderreihen vieler Punkte und Linien zeichnen (= plotten). Ein Quadrat läßt sich mit folgendem Befehl zeichnen: DRAW 1,0,0 TO 100,0 TO 100,100 TO 0,100 TO 0,0. Dadurch werden die vier Eckpunkte des Quadrats mit vier Linien verbunden (wobei jeder Eckpunkt des Quadrats geplottet wird). Aber es geht noch einfacher mit dem Befehl BOX.

DIE BOX-ANWEISUNG

```
*****
```

Dank der COMMODORE-116-Anweisung BOX wird das Zeichnen von Quadraten oder rechteckigen Figuren noch einfacher. Mit der Anweisung BOX werden die zwei gegenüberliegenden Eckpunkte festgelegt. Das obige Beispiel kann demnach auch so gezeichnet werden: BOX 1,0,0,100,100.

Die erste Zahl bestimmt wieder Zeichnen oder Löschen, und die anschließenden vier Zahlen stellen die Koordinaten der gegenüberliegenden Quadrat-Eckpunkte dar. Für die linke, obere Ecke 0,0 und für die rechte, untere Ecke 100,100 (nahe der Bildschirmmitte).

Mit Hilfe der BOX-Anweisung läßt sich jedes Rechteck zeichnen, indem nur die Eckpunkte geändert werden. Auch gedreht kann diese 'BOX' werden. Dafür ist nach der letzten Koordinate noch der Drehwinkel (in Grad) einzugeben: BOX 1,50,50,100,100,45. Mit diesem Befehl wird die 'BOX' um 45 Grad im Uhrzeigersinn gedreht dargestellt, so daß die Form eines Rhombus entsteht.

Statt der Umrißzeichnung läßt sich auch eine volle Fläche darstellen. Dies geschieht mit einer weiteren Ergänzung: nach der Winkelangabe folgt ein zweites Komma und die Zahl '1'. Eine Quadratfläche etwa in der Bildschirmmitte wird dann so erzeugt: BOX 1,100,50,220,150,,1.

WICHTIG: Auch wenn die Fläche nicht gedreht werden soll, ist dieses zweite Komma wichtig. Der Computer interpretiert dann den fehlenden Winkelwert als Unterlassung (DEFAULT = 0), und deutet dies in der Folge als eine weitere Instruktion. Fehlt das zweite Komma, so wird die Zahl '1' am Befehlsende als Drehwinkel von 1 Grad ausgelegt und die Figur um diesen Wert gedreht - und nicht gefüllt.

Einige typische Beispiele von BOX-Anweisungen:

Befehlsfolge	Ergebnis
BOX ein, Spalte 1, Zeile 1, Spalte 2, Zeile 2	Umriss
BOX ein, Spalte 1, Zeile 1, Spalte 2, Zeile 2, Winkel	Gedreht
BOX ein, Spalte 1, Zeile 1, Spalte 2, Zeile 2, Füllen	Fläche
BOX aus, Spalte 1, Zeile 1, Spalte 2, Zeile 2, Winkel, Füllen	Gelöscht

Spalte 1, Zeile 1, usw. sind Bildschirmpositionen der gewünschten Eckpunkte. Spalte und Zeile 1 ist die obere, linke Ecke der 'BOX', und Spalte und Zeile 2 die untere, rechte Ecke.

Nachstehendes Programm verdeutlicht die BOX-Anweisung (Zeile 60):

```

10 COLOR 0,1
20 COLOR 1,2
30 GRAPHIC 1,1
40 A = RND(1)*20+10
50 FOR L=0 TO 359 STEP A
60 BOX 1,100,50,220,150,L
70 NEXT L
80 FOR L=1 TO 2000: NEXT L
90 GRAPHIC 0,1

```

Mit Eingabe von RUN und Taste <Return> wird das Programm gestartet.

KREISE ZEICHNEN *****

Der COMMODORE 116 kennt auch einen Befehl, um Kreise zeichnen zu können. Ähnlich der Anweisung BOX läßt sich die Form in ein Oval oder eine Ellipse verändern. Die nicht kreisförmigen Figuren können auch gedreht werden. Auch Kreissegmente lassen sich auf diese Weise zeichnen.

DIE CIRCLE-ANWEISUNG *****

Die Standardform der Anweisung CIRCLE lautet:

Befehl	Ergebnis
CIRCLE ein, Mittelpkt Spalte, Mittelpkt Zeile, Radius	KREIS/OVAL
CIRCLE ein, Mtlpt Splte, Mtlpt Zle, Breite, Höhe	KREIS/OVAL
CIRCLE ein, Mtlpt Splte, Mtlpt Zle, Brte, Hhe, Afg, End	KREIS-SEGM
CIRCLE ein, Mtlpt Splte, Mtlpt Zle, Brte, Hhe,,,Winkel	KREIS/OVAL
CIRCLE ein, Mtlpt Splte, Mtlpt Zle, Brte, Hhe,,,,PktWkl	VIEL-ECK

Mit dem folgenden Befehl zeichnen Sie einen Kreis in Bildschirmmitte: CIRCLE 1,160,100,50. Die Koordinaten des Kreismittelpunktes liegen in Spalte 160 und Zeile 100, bei einem Radius von 50. Auf manchen Fernsehern erscheint das Ergebnis möglicherweise nicht als Kreis sondern oval, weil die Bildschirmpunkte (DOTS) höher als breit sind. Um einen 'echten' Kreis zu zeichnen, muß dem Computer in diesem Fall eine Zusatzzahl für die Höhe gegeben werden, die etwas geringer als die der Breite ist. In diesem Fall wird die vierte Zahl nämlich nicht als Radius sondern als Breite erkannt. CIRCLE 1,160,100,50,42.

Der COMMODEORE 116 kann mit dem Befehl CIRCLE auch Quadrate, Dreiecke oder andere Vielecke zeichnen. Die Zusatzangabe hierfür ist das Winkelmaß auf einem Kreis zwischen den Eckpunkten: CIRCLE 1,160,100,50,,,,120. Mit diesem Befehl entsteht ein Dreieck, dessen Winkel je 120 Grad betragen.

Erinnern wir uns, daß im CIRCLE-Befehl die anstelle der (weggelassenen) Zahlenwerte eingesetzten Kommata den Computer veranlassen, stattdessen Standardwerte einzusetzen. Die Formel für den Mittelpunktswinkel eines regelmäßigen Vielecks mit n Seiten lautet: $360/n$.

Nachfolgend ein kurzes Programm, um Vielecke zu zeichnen:

```
5 DO
10 GRAPHIC 2,1
20 INPUT "WIE VIELE SEITEN"; A
30 IF A<2 OR A>100 THEN PRINT "BITTE NICHT UEBERTREIBEN !": GOTO20
40 CIRCLE 1,160,80,40,40,,,,360/A
50 INPUT "WEITER J/N";C$
60 LOOP UNTIL C$="N"
70 GRAPHIC 0:END
```

Es steht Ihnen frei, statt eines Kreises auch nur ein Kreis-segment zu zeichnen. Die CIRCLE-Anweisung akzeptiert auch Anfangs- und Endpunkte in Grad, wenn sie anschließend an die Höhenzahl eingegeben werden. Mit dem Befehl: CIRCLE 1,160,100,50,42,90,180 wird nur das rechte untere Segment eines Kreises gezeichnet.

Ein Oval läßt sich drehen, wenn hinter dem eigentlichen CIRCLE-Befehl der Winkel (im Uhrzeigersinn betrachtet) der Drehung eingegeben wird: CIRCLE 1,160,100,100,20,,,30.

Und nun ein Programm, in dem die CIRCLE-Anweisung für einen interessanten Effekt genutzt wird. Dazu muß jedoch vorher der Speicher mit dem Befehl NEW (und Taste <Return>) gelöscht werden.

```
10 COLOR 0,1
20 COLOR 1,2
30 GRAPHIC 1,1
40 A = RND(1)*20+10
50 FOR L=0 TO 359 STEP A
60 CIRCLE 1,160,100,80,40,,,L
70 NEXT L
80 FOR L=1 TO 2000: NEXT L
90 GRAPHIC 0,1
```


DIE PAINT-ANWEISUNG

Besteht der Wunsch, von einem Kreis oder einer sonstigen geometrischen Figur nicht nur die Umrisse zu zeichnen, sondern die Fläche auch auszumalen, dann steht dafür die Anweisung PAINT zur Verfügung. Die PAINT-Anweisung füllt jede geschlossene Fläche bis an die Umrandung aus. Fehlen diese Linien, dann erfolgt die Füllung bis zum Bildschirmrand.

Die BOX-Anweisung kennt ebenfalls ein Füllkommando, mit dem Kästchen und Rechtecke eingefärbt werden können. Für alle sonstigen Figuren, die auf dem Bildschirm mit anderen Befehlen nicht gefüllt werden können, steht die Anweisung PAINT zur Verfügung. Sie füllt diejenige geschlossene Fläche aus, in der sich die angegebenen Koordinaten befinden.

Um zu sehen, wie die Anweisung PAINT u.a. einsetzbar ist, geben Sie zum vorhergegangenen Programm noch diese Zeile ein:

```
75 PAINT 1,160,100
```

```
*****  
* 6.7 MEHRFARBIGE GRAFIK *  
*****
```

Mit der Hi-Res-Grafik des COMMODORE 116 läßt sich jeder einzelne Punkt - auch 'PIXEL' genannt - auf dem Bildschirm ansprechen und steuern. Wir konnten aber auch feststellen, daß die Farben an den Stoßstellen sich verwaschen. Die meisten Hi-Res-Programme lassen überhaupt nur den Einsatz von zwei Farben zu.

Werden mehr Farben gewünscht, so verfügt der COMMODORE 116 über einen MEHRFARBEN-Befehl. Im Gegensatz zur Hi-Res-Grafik stehen im MEHRFARBEN-Modus nur die Hälfte der Punkte pro Zeile zur Verfügung. Die Punkte selbst sind doppelt so breit wie im Hi-Res-Modus; daher nur 160 Punkte pro Punktzeile - bei gleichbleibend 200 Punktzeilen. Für die MEHRFARBEN-Darstellung muß also eine etwas geringere Auflösung in Kauf genommen werden.

Zum Verständnis des MEHRFARBEN-Befehls betrachten wir nochmals den GRAFIK-Modus (Kapitel 6.3). Hier wurden bereits beide Varianten erwähnt: MEHRFARBEN-Darstellung ohne Text benötigt den Befehl GRAPHIC 3 und, mit fünf Zeilen Text kombiniert, den Befehl GRAPHIC 4.

Ebenfalls im Kapitel 6.3 wurden beim Bereich des COLOR-Befehls die Mehrfarben 1 und 2 (COLOR 2 und COLOR 3) erwähnt. Mit diesen Bereichen stehen zwei weitere Farben zur Verfügung. Jede dieser drei Farben kann eingesetzt werden: 1. die Text-Farbe, 2. die eine Extra-Farbe und 3. die andere Extra-Farbe. Diese Farben verwaschen nicht an den Stoßstellen des Bildschirms, wie dies in einigen beschriebenen Programmen des Hi-Res Modus der Fall war.

Mit dem nächsten Programm-Beispiel setzen Sie MEHRFARB-Grafik ein, wobei ein gewisser 'Leuchtreklame'-Effekt auftreten wird:

NEW

eingeben und Taste <Return>.

Mit der <STOP>-Taste kann das Programm
beendet werden. Durch Zeile 40 wird in
diesem Fall nach Zeile 200 gesprungen.

```
10 COLOR 0,1
20 GRAPHIC 3,1
30 COLOR 3,1
40 TRAP 200
50 DRAW 3,10,10 TO 10,100: DRAW 3,10,55 TO 30, 55
60 DRAW 3,30,10 TO 30,100: DRAW 3,50,10 TO 80, 10
70 DRAW 3,65,10 TO 65,100: DRAW 3,50,100 TO 80,100
80 FOR L=0 TO 7
90 COLOR 3,2,L
100 FOR M=1 TO 100: NEXT M
110 NEXT L
120 COLOR 3,1
130 FOR M=1 TO 100: NEXT M
140 GOTO 80
200 GRAPHIC 0: COLOR 1,1,7
```

Den Mehrfarben-Bereich 3 (Befehl COLOR 3) verfügt über eine weitere Fähigkeit, die keiner der anderen Bereiche besitzt. Haben Sie einmal Bildschirmbereiche mit dem Befehl COLOR 3 gezeichnet, dann kann diese Farbe an jeder Bildschirmstelle, auf der sie auftaucht, mit der COLOR-Anweisung verändert werden.

Legen Sie z.B. die Farbe mit dem Befehl COLOR 3,5 fest und zeichnen in der Folge mit dieser Farbe, so erscheint die Grafik in Purpur. Wechseln Sie nun per Befehl COLOR 3,6 die Farbe, so wechseln schlagartig alle Purpur eingefärbten Flächen in die neue Farbe Grün. Mit keinem der anderen Bereiche ist dies möglich.

T E I L 6

COMMODORE 116