

```

*****
*                                     *
*   T   E   I   L       7           *
*                                     *
*****
*                                     *
*   T   Ö   N   E           *
*                                     *
*   U   N   D               *
*                                     *
*   M   U   S   I   K       *
*                                     *
*****

```

```

*   EINLEITUNG
*   *****
*   LAUTSTÄRKEREGELUNG (VOL)
*   *****
*   TONERZEUGUNG (SOUND)
*   *****
*   KLANGEFFEKTE ERZEUGEN
*   *****

```

```
*****  
* 7.1 EINLEITUNG *  
*****
```

Zur Einleitung ein kleines Musikprogramm. Nachdem Sie es in den COMMODORE 116 eingetippt und mit 'RUN' gestartet haben, wird ein Fragezeichen am Bildschirm erscheinen. In diese INPUT-Abfrage können Sie, ganz nach Ihrem Gutdünken, die Zahlen von 1 bis 1015 eingeben - und das Programm per Taste <Return> weiterlaufen lassen. Beendet wird das Programm, indem Sie den Wert Null ('0') eingeben.

```
10 VOL 8  
20 DO  
30 INPUT X  
35 IF X>1015 OR X<0 THEN PRINT "BITTE '0' bis '1015'": GOTO 30  
40 SOUND 1, X, 10  
50 LOOP UNTIL X=0
```

Einzelne Noten werden auf dem COMMODORE 116 so erzeugt:

1. VOL 8 eintippen und Taste <Return> drücken
2. SOUND 1,266,50 eintippen und Taste <Return> drücken.

Der programmierte Ton erklingt etwa für eine Sekunde. Sehen Sie diese Note als den allerersten Schritt zu einer Symphonie, nach dem Motto: erste Note im Computer, 3 500 weitere noch zu komponieren.

Wenn Sie nichts hörten, dann ist der Lautstärkeregler des Fernsehgeräts oder Monitors nicht genügend weit aufgedreht. Holen Sie dies nach und starten Sie den Zweizeiler nochmals. Er umfaßt alles, um mit dem COMMODORE 116 Musik zu machen. Beide Befehle sind leicht zu verstehen und noch einfacher anzuwenden.

```
*****
* 7.2 LAUTSTÄRKEREGELUNG (VOL) *
*****
```

Mit dem Befehl VOL (= VOLUME) wird die Lautstärke jedes vom COMMODORE 116 erzeugten Tones gesteuert. Der Befehl funktioniert ähnlich dem Lautstärkeknopf: Folgt eine Null ('0') dem VOL-Befehl (VOL 0), dann bewirkt dies Lautstärke 0. Maximum wird mit der Zahl 8 erreicht, und die Zwischenwerte liegen von 1 bis 7.

Setzen Sie zur Übung in das vorangegangene Beispiel unterschiedliche Werte beim VOL-Befehl ein. Je größer die Zahl, desto lauter der Ton.

```
*****
* 7.3 TONERZEUGUNG (SOUND) *
*****
```

Im SOUND-Befehl sind alle übrigen zur Tonerzeugung erforderlichen Informationen für den COMMODORE 116 enthalten. Diese drei Zahlen beschreiben eine Note wie folgt:

SOUND (Stimmen-Nr.), (Noten-Wert), (Klang-Dauer)

Die erste Zahl im SOUND-Befehl betrifft die Stimme. Man kann Stimme 1, 2 oder 3 wählen. Im COMMODORE 116 sind zwei unabhängige Tongeneratoren eingebaut. Tongenerator 1 erzeugt nur Töne, während Tongenerator 2 entweder Töne oder Rauschen erzeugen kann.

Stimme 1: Tongenerator 1 erzeugt Töne. Wird mit SOUND 1 aufgerufen.


Stimme 2: Tongenerator 2 erzeugt Töne. Verhält sich wie Stimme 1, wenn die Anweisung SOUND 2 lautet.

Stimme 3: Wünschen Sie statt eines Tons Geräusche, wie z.B. Regen oder Donnergerollen, dann tippen Sie SOUND 3, Tongenerator 2 erzeugt Rauschen.

Die zweite Zahl in der SOUND-Anweisung steuert den Notenwert (die Tonhöhe der Note). Dieser Wert kann sinnvollerweise zwischen Null ('0') und 1015 liegen. Der COMMODORE 116 wird dadurch in der Tonhöhe gesteuert. Je höher die Zahl, umso höher der Ton. Die höchsten Werte (ab 1015) sind für das menschliche Ohr kaum noch hörbar.

WICHTIG: Bei Stimme 3 ergeben Werte zwischen 600 und 940 eine Art weißes Rauschen. Um interessante Klangeffekte zu erzeugen, können Register-Werte auch außerhalb dieses Bereichs eingesetzt werden.

Einige Notenbeispiele zum besseren Verständnis. Eine vollständige Tabelle finden Sie im Anhang des Handbuchs.



NOTE:	a	h	c	d	e	f	g
WERT:	770	798	810	834	854	864	881
FREQUENZ: (in Hz)	440.4	494.8	522.7	588.7	658	699	782.2

Versuchen Sie folgendes Programm:

```
NEW                               eingeben und Taste <Return>.
10 VOL 7  <----- Lautstärke '7'
20 X = 0
30 DO
40 SOUND 1,X,5  <----- Stimme, Notenwert u. Klangdauer
50 X = X + 5
60 LOOP UNTIL X = 1015
70 VOL 0  <----- Lautstärke '0' (= AUS)
80 END
```

Dieses Programm wird Sie von den musikalischen Fähigkeiten des COMMODE 116 sicherlich überzeugt haben.

Die dritte Zahl im SOUND-Befehl steuert die Klangdauer (duration) der Note in einem Bereich von 0 bis 65 535 (mal 1/50 Sekunden). Das bedeutet, daß eine Klangdauer von 50 etwa einer Sekunde entspricht. Über den Daumen gepeilt bedeutet eine umso größere Zahl eine entsprechend längere Klangdauer. So erreichen Sie mit der Zahl 65535 eine Klangdauer von über 20 Minuten. Mit dem Wert Null ('0') können Sie jeden Ton ausschalten.

```
*****
* 7.4 KLANGEFFEKTE ERZEUGEN *
*****
```

Klangeffekte lassen sich sowohl mit Musiktönen wie mit Geräusch erzeugen. Die Verbindung einfacher BASIC-Programme mit SOUND-Befehlen kann ungewöhnliche und unterhaltsame Effekte bringen. So kann (wie im nachstehenden Beispiel gezeigt) die FOR ... NEXT ... STEP ...-Schleife geschickt für Klangeffekte genutzt werden.

Mit diesem Befehl wird eine Schleife gebildet; und jedesmal wenn der Computer den Befehl FOR liest, wird die Zählervariable (in unserem Beispiel 'S') verändert. Mit Ausführung des Befehls NEXT wird zum Befehl FOR zurückgesprungen. In der verwendeten FOR...NEXT-Schleife steht ein negativer STEP Befehl, um von der höheren Zahl herunterzuzählen (jedesmal um 25).

```
10 VOL 8
20 FOR S=1000 TO 700 STEP -25

30 SOUND 1, S, 1
40 NEXT S
```

Lautstärke auf 8.
Bildet Schleife, um 25 abwärts zählend.

Geben Sie den Befehl RUN in Verbindung mit der Taste <Return> ein und lauschen Sie den Klangeffekten. Programmbestimmend ist die Zeile 20, in der im Zahlenbereich 1000 bis 700 die gesamte Skala in 25er Schritten hinuntergespielt wird. Zeile 30 entscheidet über die Dauer jedes Tons - in diesem Fall von je 1/50 Sekunde. Experimentieren Sie mit den unterschiedlichsten Zahlenwerten bei Klang und Tondauer, womit auch Ihnen einige bemerkenswerte Tonkombinationen gelingen werden.

Wird bei der Stimm-Auswahl der Wert 3 (SOUND 3,...) gewählt, dann wird vom COMMODORE 116 Rauschen erzeugt. Damit lassen sich wie bei der Tonerzeugung die unterschiedlichsten Klangeffekte erzielen. Nachstehendes Programm benützt den SOUND 3,...-Befehl (Stimme 3), um das Geräusch eines Sturmes nachzubilden.

NEW

eingeben und Taste <Return>.

```

10 VOL 2 <----- Lautstärke '2'
20 R=INT(RND(0)*10)+1 <-- Zufallszahl 'R'
30 FOR X=1 TO R
40 SOUND 3, 600+30*X, 10
50 NEXT X
60 FOR X= R TO 1 STEP -1
70 SOUND 3, 600+30*X, 10
80 NEXT X
90 T=INT(RND(0)*100)+30
100 SOUND 3, 600, T
110 GOTO 20

```

Die Zeilen 30 und 60 bilden je eine FOR...NEXT-Schleife für den Noten-Wert (Frequenz), wobei die erste ansteigende und die zweite absteigende Frequenzwerte - jeweils basierend auf den Zufallszahlen der Zeile 20 - erzeugt. Wichtig für unser Programmbeispiel ist die wechselnde Tonhöhe, da Gewitterstürme mit peitschenden Sturmböen dahergehen.

In den Zeilen 40 und 70 werden per SOUND-Befehl die Geräusche erzeugt, und in den Zeilen 90 und 100 werden mit zufälligen Verzögerungswerten die absolut unregelmäßigen Zeitabstände zwischen dem Heulen eines Gewittersturms nachempfunden.

Außerdem wird in Zeile 90 eine weitere Zufallszahl erzeugt, mit dem die Dauer des SOUND-Befehls in Zeile 100 bestimmt wird. Dieser SOUND-Befehl bleibt auf konstanter Tonhöhe und bildet als Kontrast zum Auf- und Abschwellen des Sturms ein konstantes Hintergrundgeräusch.

Das klingt natürlich alles recht kompliziert und ausschließlich für Programmierprofis machbar. In Wirklichkeit ist nur ein wenig Experimentierfreude notwendig, wobei unterschiedliche Werte eingesetzt werden und man beobachtet, was dabei herauskommt. Geräusche zu erzeugen ist eine Herausforderung, bei der versucht wird, die richtigen Werte für einen gewünschten Effekt zu finden.

WIR MACHEN MUSIK

Auch wenn Sie (noch) nicht alles verstehen, was in allen diesen Programmen abläuft, so tippen Sie die Programme dennoch ein und beobachten ihre Wirkung.

Das folgende Programm simuliert über die Tasten <1> bis <8> eine Klaviertastatur.

```
5 SCNCLR
7 PRINT"SPIELEN SIE MIT DEN TASTEN 1..8!"
8 PRINT"ENDE:LEERTASTE"
10 FOR X=1 TO 8: READ N(X): NEXT X
20 VOL 7
30 DO
40 GET A$: IF A$="" THEN 40
50 A=ASC(A$): IF A<49 OR A>56 THEN 90
60 N=A-48
70 SOUND 1, N(N), 5
80 COLOR 0, N, 3
90 LOOP UNTIL A=32
100 VOL 0: COLOR 4, 2, 7
110 DATA 169, 262, 345, 383, 453, 516, 571, 596
```

Spielen Sie mit den Tasten <1> bis <8> die Noten. Mit jeder anderen Note wechselt auch die Bildschirm-Randfarbe. Wollen Sie beenden, so genügt ein Druck auf die LEER-Taste.

Nachdem der COMMODORE 116 auch als Klavier zur Verfügung steht, wollen Sie vielleicht ein bekanntes Liedchen spielen. Ein bekanntes, ja klassisches Beispiel ist mit den zu drückenden Zifferntasten nachstehend abgedruckt:

Tasten der Reihe nach (im Takt) drücken:

```
1 1 5 5 6 6 5
4 4 3 3 2 2 1
5 5 4 4 3 3 2
5 5 4 4 3 3 2
1 1 5 5 6 6 5
4 4 3 3 2 2 1
```

Die einzelnen zu spielenden Noten können auch in DATA-Werten (z.B. für die Variablen X und Y) abgelegt und durch das Programm aufgerufen (READ) werden, wobei sich ihr Wert bei jeder Schleife ändert. Im nächsten Programm sind diese DATA-Werte paarweise zu sehen. Der erste Wert bildet stets den Notenwert für den SOUND-Befehl und der zweite Wert erzeugt im SOUND-Befehl die Klangdauer.

'Alle Vögel sind schon da'

```
10 VOL8
20 DO
30 READX,Y
40 SOUND1,X,Y
50 LOOP UNTIL X=0
60 END
100 DATA 596,45,685,15,739,30,810,30
110 DATA 770,30,810,15,770,15,739,60
120 DATA 704,45,739,15,685,30,596,30
130 DATA 643,60,596,30
140 DATA 0,1
```

COMMODORE 116 - DIE MUSIKMASCHINE

Das letzte Programm ist etwas länger. Es zeigt den COMMODORE 116 als die 'große' Musikmaschine. Jede Taste zwischen <1> bis <9>, die Sie drücken, erscheint auch als geschriebene Note in den Notenzeilen.

```

5 GOSUB 1000
6 FOR X=1 TO 9: READ N(X): NEXT X
8 CHAR 1, 8, 1, "DIE GROSSE MUSIKMASCHINE"
10 VOL 7
20 DO
30 GET A$: IF A$="" THEN 30      zwischen den Anführungszeichen
                                keinen Zwischenraum!
35 A=ASC(A$): IF A<49 OR A>57 THEN 50
36 N= A - 48
40 SOUND 1, N(N), 4
45 GSHAPE N$, 150, 8* (6+(9-N)), 4
46 FOR Z=1 TO 50: NEXT Z
47 GSHAPE N$, 150, 8* (6+(9-N)), 4
50 LOOP UNTIL A=32
55 VOL 0: GRAPHIC 0: SCNCLR
60 END
100 DATA 345, 383, 453, 516, 571, 596, 643, 685, 704
1000 GRAPHIC 1,1
1010 FOR Y=60 TO 124 STEP 16
1020 DRAW 1, 100, Y TO 200, Y
1030 NEXT Y
1040 A$="FEDCHAGFE"
1050 FOR X=1 TO 9: C=13
1060 IF INT(X/2)= X/2 THEN C=14
1070 CHAR 1, C, X+6, MID$ (A$, X, 1), 0
1075 CHAR 1, C+10, X+6, RIGHT$ (STR$ (10-X), 1)
1080 NEXT X
1090 FOR X=1 TO 8: FOR Y=11 TO 16: DRAW 1, X, Y: NEXT Y, X
1100 Y=1: X=8: DRAW 1, 8, 16 TO X, Y
1110 SSHAPE N$, 1, 1, 8, 16
1120 GSHAPE N$, 1, 1, 4
1130 RETURN

```

Wie Sie feststellen konnten, können Musik und Töne sowohl zur Programmuntermalung genutzt werden oder auch Hauptpunkt eines Programms sein. Die gebrachten Beispiele dieses Kapitels konnten Ihnen nur einen Vorgeschmack der Möglichkeiten des COMMODORE 116 bieten. Scheuen Sie nicht neue Töne und Geräusche zu erproben, und komponieren Sie Ihr eigenes Meisterstück.

Damit schließt der Einführungsteil des COMMODORE 116 Handbuchs. Aufgabe dieses Teils war es, Ihnen einen Vorgeschmack dessen zu vermitteln, was in einem COMMODORE 116 steckt. Sie müssen es weiter versuchen mit neuen Programmen, auf die Ergebnisse achten und dabei mehr über Ihren Computer lernen, damit Sie Freude an Ihren Fortschritten haben. Und es gibt noch sooo viel zu programmieren.

Dieses Handbuch kann erste Schritte in der Computersprache BASIC vermitteln, stellt jedoch in keiner Weise ein Studier- oder BASIC-Lehrbuch dar. Jedoch gibt das nachfolgende BASIC-Lexikon eine Gesamtaufstellung ALLER BASIC-Befehle des COMMODORE 116, einschließlich Erklärungen und Beispielen. In vielen der vorangegangenen Programmbeispiele haben Sie vielleicht nicht jeden Schritt nachvollziehen können. Zu Ihrer Weiterbildung empfehlen wir Ihnen die Lektüre spezieller BASIC-Lehrbücher. Die COMMODORE-Sachbuchreihe bietet ein besonders interessantes Spektrum an weiterführender Literatur. In diesen Büchern ist das Hauptaugenmerk auf das 'WIE' und 'WARUM' der Programmierkunst gelegt, und es werden die ganz geheimen und die weniger geheimen Tricks verraten, wie das meiste aus Ihrem Computer herauszuholen ist.

Befürchten Sie nicht, nach der Lektüre dieses Handbuchs in Zukunft allein und ohne weitere Unterstützung zu sein. Es gibt eine Fülle einschlägiger Fachzeitschriften, USER-Gruppen und Vereine, die Ihnen gerne zur Seite stehen werden.